' Error Handling 5 - © 1993-2003 Aiman Hanna

' This program illustrates more of error handling. In specific the

' program illustrates more of the "Resume" statement.

' Key Points:

'       1) "Resume"

'       2) "Resume Next"

Option Explicit

Private Sub cmdDivide_Click()

    ' This sub should allow the user to run 5 divisions. If an error

    ' occurs because the user enters a bad value, or nothing, the program

    ' will show the error and goes back to where the error occurs (asks the

    ' user again to enter a good value). If however the user enters a zero as

    ' a second value, the program will count this failed attempt as one of the five

    ' allowable attempts and continue to the next iteration of the loop, if any.

    ' In this function we will check against two possible errors.

    ' The first possible error is that there is nothing entered

    ' by the user or if the entry is not an integer value. The

    ' second possible error is that the user will indeed enters

    ' two integers values but the second value is zero which

    ' we can not divide by. We will have two "On Error" statements

    ' to avoid both of these potential errors.

    Dim val1%, val2%, counter%

    For counter = 1 To 5

        txtCounter.Text = counter

        On Error GoTo InvalidInputHandler   ' set trap

        ' The above error trap will execute if an error occurs now. That is when

        ' the user does not enter any values in one (or both) of the textboxes

        ' or if the user enters invalid values

        val1 = InputBox("First Value", "Enter first value")

        val2 = InputBox("Second Value", "Enter second value")

       ' If this line is reached then there must have entered two

       ' integer values. Now must validate the second value is not zero.

       ' If the second value is zero then the second On Error statement

       ' will fire/execute.

       On Error GoTo DivideByZeroHandler

        txtDivVal.Text = val1 \ val2

    Next counter

    Exit Sub    ' exit the sub so the error handler does not execute

InvalidInputHandler:        ' label for error handler

   Call MsgBox("Invalid Input. You must enter two integers.", vbOKOnly, "Invalid Input")

   Resume    ' go back to the same line that caused the problem and execute it again

DivideByZeroHandler:        ' label for error handler

   Call MsgBox("Attempt to divide by zero. Second value cannot be zero.", vbOKOnly, "Division By Zero Error")

   Resume Next   ' go to the next line to the one that caused the problem and execute it

End Sub

Private Sub cmdExit_Click()

    End

End Sub

